

Taulukot

Yksiulotteisen taulukon määrittely

```
real x(10)
real :: x(10)
real, dimension(10) :: x
```

Indeksin alarajan oletusarvo on 1. Voidaan muuttaa määrittelyssä:

```
real x(0:10)
real, dimension(-10:10) :: y
```

Myös taulukko voidaan alustaa määrittelyn yhteydessä:

```
real, dimension(3) :: x = (/ 0.0, 0.5, 1.0 /)
real, dimension(-10:10) :: y = 0.0
```

Useampiulotteiset taulukot:

```
real y(2,3)
real, dimension(-1:1,0:100) :: y

y(1,1)  y(1,2)  y(1,3)
y(2,1)  y(2,2)  y(2,3)
```

Taulukoiden talletustapa poikkeaa muista kielistä:

Fortranissa 2-ulotteinen taulukko talletetaan sarakkeittain:

```
real y(0:1,0:2)

y(0,0) y(0,1) y(0,2)
y(1,0) y(1,1) y(1,2)
```

Tämän alkiot ovat muistissa järjestyksessä

```
y(0,0) y(1,0) y(0,1) y(1,1) y(0,2) y(1,2)
```

C:n taulukko y[2][3] talletetaan riveittäin

```
y[0][0] y[0][1] y[0][2] y[1][0] y[1][1] y[1][2]
```

Taulukko-operaatiot

Laskutoimitus voi kohdistua kokonaiseen taulukkoon:

```
real, dimension (100) :: x, y, z, u
x=y+z
u=x*z
```

Operaattorit kohdistuvat vastinalkioihin; * ei ole matriisitulo.

Lausekkeessa esiintyvien taulukoiden oltava yhteensopivia.

Sijoituslauseen molemmilla puolilla oltava yhteensopivat taulukot.

Poikkeus: skalaari on yhteensopiva kaikkien taulukoiden kanssa:

```
real x(100), y(100), a
y=1.0
x=y+a
```

Varusfunktiot voivat kohdistua kokonaiseen taulukkoon

```
real x(100), y(100)
x=0.1*(/ (i, i=1,100) /)
y=sqrt(x)
```

Taulukoihin liittyviä varusfunktioita

```
real x(100), x0, x1
integer, dimension(1):: k0, k1, i0, i1

k0 = lbound(x) ! indeksin alaraja
k1 = ubound(x) !      ylaraja

x0 = minval(x) ! pienin arvo
x1 = maxval(x) ! suurin arvo

i0 = minloc(x) ! pienimman arvon indeksi
i1 = maxloc(x) ! suurimman arvon indeksi

real x(2,3)
integer lo(2), hi(2), i1(2)
x(1,:) = (/ 1, 2, 3 /)
x(2,:) = (/ 2, 5, 1 /)
lo=lbound(x) ! lo = (/ 1, 1/)
hi=ubound(x) ! hi = (/ 2, 3/)
i1=maxloc(x) ! i1 = (/ 2, 2/)

real, dimension (10:14) :: &
    x=(/ 0.0, 1.0, 5.0, 2.0, 1.0 /)
integer, dimension(1):: l, n, i

l = lbound(x) ! l = (/ 10 /)
n = ubound(x) ! n = (/ 14 /)
i = maxloc(x) ! i = (/ 3 /)
```

Vektori- ja matriisioperaatioita:

```
real, dimension (10,10) :: a, b, c
real, dimension(10) :: v1, v2
real z, s, p

s = sum(a)          ! alkioiden summa
p = product(v1)     ! alkioiden tulo
z = dot_product(v1, v2) ! skalaaritulo
c = matmul (a, b)   ! matriisitulo
```

Taulukkosektiot

Taulukon paikalla voi olla myös taulukon osa eli taulukkosektio.

```
real a(100), b(100), c(100), d(10,100)
integer i,j

b(1:10) = c(51:60)
a=d(1,:)
c=a(100:1:-1)+b

i=4 ; j=20
d(1:3,1:5) = d(i:i+4:2, j:j+4)

d(:,1) = d(5, 10:19)
```

Taulukkosektio voi olla mielivaltainen suorakulmainen, tasavälinen hila.

Sijoituslauseen oikea puoli lasketaan aina kokonaan ennen sijoitusta.

```
integer a(3)
a= (/ 0, 1, 2 /)
a (2:3) = a (1:2) ! a=(/ 0, 0, 1 /)
```

Tämä toimii eri tavalla kuin silmukka

```
do i=2,3
  a(i) = a (i-1)
end do
```

Taulukoiden käsittelyyn where-lause:

```
where (x /= 0.0) y=1/x
```

```
where (x > 0.0)  
  y=1/x  
  z=log(x)  
end where
```

```
where (x > 0.0)  
  y=1/x  
  z=log(x)  
elsewhere  
  y=0.0  
  z=0.0  
end where
```